Canceling Long Operations

Present article tries to explain how to write commands for aborting or canceling long operations. Depending on many situations, long operations can take from a few seconds to some hours to complete, and providing a way to cancel this operations is an important feature in any application.

Besides explaining how to cancel operations, this article presents code to validate numeric text boxes and disable all form controls.

Developing an application with canceling/aborting feature is not very different from normal applications, requiring some attention on the following points:


1 – Developer must provide a "Cancel" button to users: many applications make use of dialog boxes or user forms, and the most common solution is to use the "Cancel" button. In this circumstance, the "Cancel" button serves to close a user form, or to cancel long operations.


2 – Developer must provide users a way to press the "cancel" button during long operations: By default, when code is running, users can't click a "cancel" button, and developers must carefully employ DoEvents functions to correct this problem.


3 – When employing DoEvents functions, users have access to form controls and can editing all form controls, introducing errors to applications. Facing this problem, developers have to disable all form controls, except "cancel" button.

With all these aspects in attention, I will explain how my small application works. After opening the spreadsheet, click "Open Form" button to open the following user form, fill numeric controls and press the Run button.

After pressing Run button, all form controls except Cancel button become disabled.

When users click on Cancel button, application requires users to confirm before canceling the operation.

[image: image3.png]Microsoft Excel

@ Do You Want To Cancel Current Operation?





How to code this application? First define two module level variables of type Boolean to save information how the application is working:


	Variable Name
	Function

	Dim bolCodeIsRunning As Boolean
	True when main code is running, False in all other conditions

	Dim bolCancelCode As Boolean
	True when user press Cancel button, False in all other conditions


As previously mentioned, the second step is to code the Cancel button to serve two different purposes. If code is running, asks users to confirm before canceling operation. Otherwise, code is not running, and user click cancel to close user form. 

Please note that when users confirm to cancel this operation, the only thing is done is to change variable bolCancelCode to True, and is on the main code that the code stops working.

Private Sub cmdCancel_Click()

  If bolCodeIsRunning Then

    If MsgBox("Do You Want To Cancel Current Operation?", vbQuestion + vbYesNo) = vbYes Then

      bolCancelCode = True

    End If

  Else

    Unload Me

  End If

End Sub

Finally, we have to write the main code so all this works. 

Private Sub cmdRun_Click()

Dim lngCounter As Long

'Checks all text box for valid numeric values

Select Case False

  Case IsNumericAndValid(txtFrom, 0, 10000):

    MsgBox "Enter a valid Number in From box"

    Exit Sub

  Case IsNumericAndValid(txtTo, Val(txtFrom), 10000)

    MsgBox "Enter a valid Number in To box"

    Exit Sub

  Case IsNumericAndValid(txtDelay, 0, 10000)

    MsgBox "Enter a valid Number in Delay box"

    Exit Sub

End Select

'Turns variables to run mode

bolCodeIsRunning = True

DisableAllFormControls (Me)

cmdCancel.Enabled = True

For lngCounter = Val(txtFrom) To Val(txtTo)

  Application.StatusBar = "Running: " & lngCounter

  Delay (Val(txtDelay))

  DoEvents

  If bolCancelCode Then Exit For

Next lngCounter

'Turns Off Variables

Application.StatusBar = False

Unload Me

End Sub

As in any other application, after pressing Run button, is necessary to validate any information provided by users, and I used a select case and a custom function to provide all functionality to this task. This custom function checks if control is numeric and between two range of values.

Function IsNumericAndValid(strTestValue As String, dblGreaterThan As Double, dblLessThan As Double) As Boolean

IsNumericAndValid = False

If IsNumeric(strTestValue) Then

  If Val(strTestValue) > dblGreaterThan And Val(strTestValue) < dblLessThan Then

    IsNumericAndValid = True

  End If

End If

End Function

After validation, turn variable to run mode. In this small example, turn bolCodeIsRunning variable to True, disable all forms controls, and enable cancel button. As I mentioned on my first article to OzGrid newsletter about reusable code, I developed a general-purpose procedure to work with any user form. 

Sub DisableAllFormControls(frmControls As Controls)

Dim varFormControl As Variant

For Each varFormControl In frmControls

  varFormControl.Enabled = False

Next varFormControl

End Sub

DoEvents function is normally used on any cycle structure control, like For Next, While or Do loop. DoEvents purpose is " Yields execution so that the operating system can process other events. (Copied from Microsoft Excel Visual Basic Help)", and this mean that users have access to click on cancel button.

So, when users click the cancel button and after confirmation, variable bolCancelCode is changed to True. Since this application uses a cycle with a For structure, I insert a line of code to detected if user canceled the operation and to exit the For cycle:


 If bolCancelCode Then Exit For

As provided on this article, writing applications with canceling/aborting feature is not a very difficult task and is very important for long operations, like reading from and writing to Databases. Depending on application purpose and time required to complete a cycle, users may have to click several times on cancel button to cancel the operation.

During the next months, I hope to write some of my applications code at Ozgrid newsletter. For suggestions and comments, please write to software@mgmoreira.com

